
プログラムを読もう

　どのような言語を使うにせよ，プログラミングを楽しむためにはその言語の構文と文法を身につけなければならな

い．これは自然言語と同じである．しかし，エッセンスのみ取り出した構文と文法だけを暗記するような学習はあま

り楽しいものではない．今回は，Haskellを用いた簡単なプログラミングを示し，そのプログラムコードの「読み方」

をできるだけ丁寧に説明する．アルゴリズム自体はさほど難しいものではないので，Haskellで記述するとどのよう

になるか理解していただけると思う．プログラムコードが読めるようになれば，書くのはすぐにできるようになるだ

ろう．

例題 1: 二分木の列挙

　「型慣らし」にごくやさしい問題をやってみよう．

課題　末端のノードが n個の二分木を列挙せよ．

図 -1　末端が 4個の二分木

　たとえば，末端が 4個の二分木は図 -1のように 5個ある．これを末端が n個の場合に一般化しようというのであ

る．このような二分木を Haskellではどのように定義するのだろうか．Haskellでは組み込みの型以外をプログラマ

が data宣言によって定義できる．たとえば，このように定義する．

    data DumbTree = Empty | Fork DumbTree DumbTree

　これは新しいデータ型を作るときの Haskellの構文で data宣言といい，「DumbTree型は，末端ノード Emptyか

あるいは (再帰的に ) 2つの部分木 DumbTreeを含む Forkノードで構成される」と読む．このようなデータ宣言を

木 (tree)で遊ぶ

読む場合には，DumbTreeは型構成子 (type constructor)，Emptyおよび Forkはデータ構成子 (data constructor)

であることに注意する．型構成子を適用して型を得ることと，データ構成子を適用して値を得ることは別である．前

者はコンパイルの仕事で，プログラム全体にわたっての型の整合性をチェックするときのみ用いられ，後者は実行の

仕事で実際の計算法を決めるものである．

    -- trees 整数(>0)から，DumbTreeのリストへ
    trees :: Int -> [DumbTree]

    trees 1 = [Empty]

    trees n = concat [ joins ls rs | (ls,rs) <- [ lrs xs ys | (xs,ys) <- splits1 n ]]

    -- splits1 整数(>0)から，和がその数になる整数(>0)の2つ組のリストへ
    splits1 :: Int -> [(Int,Int)]

    splits1 1 = []

    splits1 n = (1,n-1) : [ (i+1,j) | (i,j) <- splits1 (n-1) ]

    -- lrs 2つの整数から，それぞれの大きさのDumbTreeのリストの2つ組へ
    lrs :: Int -> Int -> ([DumbTree],[DumbTree])

    lrs xs ys = (trees xs, trees ys)

    -- joins 2つのDumbTreeのリストから，それらをそれぞれ左右に持つDumbTreeのリストへ
    joins :: [DumbTree] -> [DumbTree] -> [DumbTree]

    joins ls rs = [ Fork l r | l <- ls, r <- rs ]

　treesが与えられた整数 (Intは Haskellに組み込みの固定長整数 )を二分木に変換する関数である．最初の行は，
treesがどのような型を持っているかの型宣言で，「treesは Int型から DumbTree型を要素とするリストへの関

数という型を持つ」と読む．型宣言中の [T ]は T 型を要素とするリストという型を示す．Haskellのリスト型は要素

の型が単一でなければならないことに注意する．2行目は自明な場合の定義である．1個の末端を持つ木は Empty

ノード 1つだけの木が唯一である．それゆえ，それだけを含むリスト [Empty]である．こちらの [Empty]は「型」

ではなく「データ」である．3行目は 1より大きい nに treesを適用したものの定義である．その右辺は，

    concat X

というかたちをしている．これの Xの部分は，

    [ joins ls rs | (ls,rs) <- Y ]

というかたちである．これはリストの内包表記と呼ばれる構文である．リストの内包表記 [ f x | x <- xs ]は

数学の { f(x)  x  X }と対応しており，Haskellでの意味は，map f xsと同等である．Yの部分はリストで，その

要素は 2つ組 (tuple)であることが分かる．このリストの内包表記は，「Yの要素である 2つ組のそれぞれ第 1要素
(ls)と第 2要素 (rs)に関数 joinsを適用した結果構成されるリスト」と読む．Yの部分は，

    [ lrs xs ys | (xs,ys) <- splits1 n ]

である．「splits1 nで構成されるのはリストの要素である 2つ組のそれぞれの前の要素 xsと後の要素 ysに関数
lrsを適用した結果構成されるリスト」と読む．

　これで treesの読み方が分かったので，今度は内側から見ていこう．splits1は与えられた整数を左右の部分

木に含まれる末端ノードの数 (1以上 )に二分割する関数である．この関数は，Intから Intの 2つ組のリストへ

の関数という型を持つ．1は 1以上の部分に分割できないので，空リスト []．1より大きい nに対しては，1に
splits1を適用してできた 2つ組のそれぞれについて最初の要素に 1を加えたリスト (ここまでがリストの内包表

記で定義されている ) の先頭に，(1,n-1)という 2つ組を付けたものである．ここで，:は Haskellでは組み込み

のリストの構成子であり，a -> [a] -> [a]という型を持つ．

　lrsは 2つの分割された数からそれぞれ左右の部分木となるべき二分木をつくる関数である．この関数の意味は

あきらかだろう．

　joinsは左側の枝になるべき部分木のリストと右側の枝になるべき部分木を組み合わせて Forkノードを構成す

る関数である．このリストの内包表記は数学の表記法 { App(l,r)  l  L  r  R }に対応している．これも意味はあ

きらかだろう．この式で構成されるリストの長さは，引数 lsの長さと引数 rsの長さの積になることに注意する．

　最後に concatは標準プレリュードで定義されている関数で，要素がリストであるようなリストに適用すると，

要素であるリストを連結し全体として 1つのリストにする関数である．concatの型は [[a]] -> [a]すなわち a

型の値を要素とするリストのリストから a型の値を要素とするリストへの関数の型を持つ．ここで，型が分かると

関数の意味がよりよく理解できることに注目してもらいたい．Haskellのプログラミングにおいては型宣言は必ずし

も必要ではないが，型宣言は関数の仕様のサマリでありプログラマの意図 (の一部 ) を表現したものと見なせる．そ

れゆえ，型宣言を書いておくことは重要である．

　さて，ここまでのコード (DumbTree，trees，splits1，lrs，および joinsの定義 )をファイル first.hs

に保存し，解釈系を使って評価してみよう．今回は解釈系に GHCに付属してくる ghciを使う．シェルからファイ

ルを指定して ghciを起動すると解釈系がファイルをロードしてユーザの入力待ちプロンプトが出る．

    % ghci first.hs

        ___         ___ _

       / _ \ /\  /\/ __(_)
      / /_\// /_/ / /  | |      GHC Interactive, version 6.4, for Haskell 98.
     / /_\\/ __  / /___| |      http://www.haskell.org/ghc/
     \____/\/ /_/\____/|_|      Type :? for help.

    Loading package base ... linking ... done.

    Compiling Main             ( first.hs, interpreted )

    Ok, modules loaded: Main

    *Main> :quit

    %

　解釈系を終了するには，解釈系のプロンプトから :quitと入力すればよい．では，定義した式を試してみよう．

    *Main> trees 4

    <interactive>:1:

         No instance for (Show DumbTree)

           arising from use of `print' at <interactive>:1

         In a 'do' expression: print it

    *Main>

　何かエラーメッセージらしきものが出ている．DumbTreeが Showクラスのインスタンスとして宣言されていな

いので印字ができないと言っているのである．Haskellの解釈系では，ある型のデータを表示するためには，その型

が Showクラスのインスタンスであることを宣言しておかなければならない．Haskellの組み込み型については標準

プレリュードで Showクラスのインスタンスとして宣言されているが，DumbTreeは新たに定義した型なので，こ

こで Showクラスのインスタンスであることを宣言する．型クラスについては回をあらためて説明する．ここで

は，DumbTree型が Showクラスのインスタンスであることを宣言する具体的な方法のみを確認しておく．これには，
DumbTree型のデータ用の Show関数を定義する．Show関数は当該の型のデータからそのデータの印字表現となる

文字列への変換関数である．この場合 Showの型は DumbTree -> Stringである．

    instance Show DumbTree where

       show Empty      = "0"

       show (Fork l r) = "(" ++ show l ++ "^" ++ show r ++ ")"

　末端ノード Emptyのときは文字列 "0"に，Forkノードのときは，まずカッコ ("(")，左の枝の印字表現，文字

列 "^"，そして右の枝の印字表現，最後にコッカ (")")を，連結したものである．++はリストを連結する演算子で

ある．Haskellにおいて文字列型 Stringは文字型のリスト [Char]の別名である．したがって，文字列はリストを

連結する演算子 ++を用いて連結することができる．このインスタンス宣言を先刻の first.hsファイルに追加しよ

う．そうしておいて，ghciの対話環境の中からこれを :loadコマンドを使ってロードし，もう一度，trees 4を

評価してみよう．

    *Main> :load first.hs

    Compiling Main             ( first.hs, iterpreted )

    Ok, modules loaded: Main.

    *Main> trees 4

    [(0^(0^(0^0))),(0^((0^0)^0)),((0^0)^(0^0)),((0^(0^0))^0),(((0^0)^0)^0)]

    *Main>

　これは，図 -1に対応しているのが分かるだろうか．

練習問題 1  trees 4の評価結果が

    [(L^(L^(L^R))),(L^((L^R)^R)),((L^R)^(L^R)),((L^(L^R))^R),(((L^R)^R)^R)]

    
と表示されるよう DumbTree用の showの定義を変更せよ．

例題 2: 切符問題

　筆者は最近はプリペイドカードなどを使うことが多く，あまり切符を購入しなくなったが，近距離の鉄道の切符に

は 4つの数字が印刷されていた．この 4つの数字と四則を使って，10を作るという遊びをやったことがあるかもし

れない 1）．ここでは以下のような問題を設定する．

課題　与えられた 1桁の数すべてと四則演算 (+，-，*，/)を使って，指定された数の作り方を示せ．

　たとえば，「1，1，9，9を使い 10を作れ」なら，(1+(1/9))*9が解答 (の 1つ ) である．ここでは，与えられ

る 1桁の数の個数は限定しない．

四則計算を表す計算木を定義する
　この問題を解くのに，与えられた数字で考えられるすべての四則計算式を構成してそれが指定された数になるかを

チェックするという方法を採用しよう．計算式は図 -2のように木で表現できる．この木 (図 -2) の形は先刻の例題

にあった木 (図 -1) と同じである．

図 -2　計算式と対応する木

　先刻の例題では木の形だけを問題にしたので，木には何の付加情報も持たせてはいなかったが，この問題で使う計

算木 (Term型 )では，末端ノードには数字 (Char型 )を，節のノードには演算子 (Char型 )を，それぞれ持たせる．

Haskellでは以下のように定義する．

    data Term = Val Char | App Char Term Term

　これは，「Term型は，Char型の値を含む Valノードか，Char型の値と，(再帰的に ) 2つの部分木 (Term型 )

を含む Appノードで構成される」と読む．

リストから計算木へ
　先刻の例題では末端ノードの数から木を構成したが，今度は末端ノードに配置される数字のリストと節ノードに配

置される演算子のリストから計算木を構成する．

    trees :: [Char] -> [Char] -> [Term]

    trees ds os = [ t | (_,t) <- [ otree os u | u <- dtrees ds] ]

    {- (_,t)のアンダースコアは2つ組の最初の項は使用しないことを示す -}

    dtrees :: [Char] -> [Term]

    dtrees [x] = [Val x]

    dtrees ds  = concat [ joins ls rs | (ls,rs) <- [ lrs xs ys | (xs,ys) <- splits1 ds ]]

    splits1 :: [Char] -> [([Char],[Char])]

    splits1 [x]    = []

    splits1 (x:xs) = ([x],xs) : [ (x:ys,zs) | (ys,zs) <- splits1 xs ]

    lrs :: [Char] -> [Char] -> ([Term],[Term])

    lrs xs ys = (dtrees xs,dtrees ys)

    joins :: [Term] -> [Term] -> [Term]

    joins ls rs = [ App '^' l r | l <- ls, r <- rs ]

    otree :: [Char] -> Term -> ([Char],Term)

    otree os (Val c)     = (os,Val c)

    otree os (App _ l r) = (os'', App o' l' r') -- 変数名(先頭文字以外)に ' を使用できる．
        where (o':os',l') = otree os  l         -- 定義の左辺にパターンを使用できる
              (os''  ,r') = otree os' r

    instance Show Term where

       show (Val c)     = [c]

       show (App o l r) = "(" ++ show l ++ [o] ++ show r ++ ")"

　ここでは見通しを良くするために，末端ノードに数字を配置するパス (dtrees)と，節ノードに演算子を配置する

パス (otree)に分けてある．dtreesは，先刻の例題の treesと比べると，引数が Intから [Char]に，返り値

が [DumbTree]から Termに代わっているが，先刻の例題の treesと同じ構造である．数字のリストを左右に分け

るのが splits1，左右の数字のリストから左右の部分木を構成するのが lrs，左右の部分木のリストから演算適用

の Appノードを合成するのが joinsである．joinsでは，演算子は仮に '^'に固定してある．

　otreeが，演算子のリストと，演算子を '^'に固定した計算木から，与えられた演算子を持つ計算木を構成する

関数である．otreeの定義の 1行目は例によって，型宣言で，「otreeは，Char型の値を要素とするリスト (演

算子のリスト )と Term型の計算木 (演算子がまだ配置されていない計算木 )とから，未配置の演算子のリストと
Term型の計算木 (演算子も配置済みの計算木 )の 2つ組への関数の型を持つ」と読む．返り値が計算木だけではな

く，未配置の演算子リストを含んでいることに注意すること．2行目は，「第 2引数の計算木が末端ノードなら，第

1引数の演算子リストと，その末端ノードを 2つ組にしたもの」と読む．3行目は，「第 2引数が演算適用ノードな

ら，os''と，演算子を o'とし，左右の部分木を l'および r'とする演算適用ノードとする」と読む．次の行の
where節 (where clause)は局所定義である．ここで，o'は otreeを元の計算木の Appノードの左部分木 lに適用

した結果の 2つ組の第 1要素のリストの先頭．l'はその第 2要素，r'はその第 1要素のリストの残りの部分 os'

に otreeを適用した結果の第 2要素．そして，os''はその結果の第 1要素である．

　Termの場合も表示するためには Showクラスのインスタンスにする必要がある．そのインスタンス宣言が最後の

3行である．ここまでで，計算木の構成部分が定義できたので実際に評価してみよう．Termデータの定義以降の関

数の定義をファイル second.hsに保存して，ghciにロードし，途中の関数の様子を見よう．

    *Main> :load second.hs

    Compiling Main             ( second.hs, iterpreted )

    Ok, modules loaded: Main.

    *Main> splits1 "0123"

    [("0","123"),("01","23"),("012","3")]

    *Main> lrs "01" "23"

    ([(0^1)],[(2^3)])

    *Main> [lrs xs ys |(xs,ys)<- splits1 "0123"]

    [([0],[(1^(2^3)),((1^2)^3)]),([(0^1)],[(2^3)]),([(0^(1^2)),((0^1)^2)],[3])]

    *Main> [joins ls rs | (ls,rs)<-[lrs xs ys |(xs,ys)<- splits1 "0123"]]

    [[(0^(1^(2^3))),(0^((1^2)^3))],[((0^1)^(2^3))],[((0^(1^2))^3),(((0^1)^2)^3)]]

    *Main> otree "+-*" (App '^' (App '^' (Val '0') (Val '1')) (App '^' (Val '2') (Val '3')))

    ("",((0+1)-(2*3)))

    *Main> trees "1234" "+-*"

    [(1+(2-(3*4))),(1+((2-3)*4)),((1+2)-(3*4)),((1+(2-3))*4),(((1+2)-3)*4)]

    *Main>

　trees "1234" "+-*"を評価すると図 -2に対応する計算木ができたことが確認できる．

計算木を評価する
　計算木を評価する関数を定義する．計算は割り算があるので，有理数で行うようにする．まず，有理数を分母と分

子の 2つ組として定義する．

    type Rat = (Int,Int)

　この type宣言は，既存の型に別名を付けるものである．計算木の評価関数 evalは以下のとおりである．

    eval :: Term -> Rat

    eval (Val x)     = ctor x

    eval (App o l r) = (ctoo o) (eval l) (eval r)

    ctor :: Char -> Rat

    ctor x = (ord x - ord '0',1)

    ctoo :: Char -> (Rat -> Rat -> Rat)

    ctoo '+' (x,y) (z,w) = (x*w+z*y,y*w)

    ctoo '-' (x,y) (z,w) = (x*w-z*y,y*w)

    ctoo '*' (x,y) (z,w) = (x*z,y*w)

    ctoo '/' (x,y) (z,w) = if z == 0 then (0,0) else (x*w,y*z)

　evalの定義はやさしいだろう．その型は「計算木 Term型から有理数 Rat型への関数の型」である．Valノード

のときは含まれている数字を有理数に変換したもの．Appノードのときは，演算子文字 oに ctooを適用して得た
Rat同士の演算関数を，左の部分木を評価した結果と右の部分木を評価した結果に適用する．

　ctorは 1桁の整数を表現する Char型のデータを有理数 Rat型に変換する．ここで使われている ordは文字の

ASCIIコードを返す関数で，標準プレリュードでは定義されておらず，標準ライブラリの Charモジュールで定義さ

れている．これを利用するにはソースコードファイルですべてのデータや関数の定義の前に import宣言によって
Charモジュールをインポートする．

    import Char

　この 1行を second.hsの先頭に追加しておこう．

可能なすべての計算木の構成
　残りは可能な計算木をすべて構成して，そのなかで指定された数になるものを選べばよい．

    ticket :: Int -> [Char] -> Term

    ticket n ds = head (filter (same n) (allterms ds))

    same :: Int -> (Term -> Bool)

    same i t = i*d == n && d /= 0

       where (n,d) = eval t

    allterms :: [Char] -> [Term]

    allterms ds = concat [ trees ns os | ns <- perm ds, os <- rperm ops4 (length ds - 1) ]

    ops4 = "+-*/"

    perm [] = [[]]

    perm xs = concat [ pm hs ts | (hs,ts) <- splits xs ]

       where pm _  []     = []

             pm hs (t:ts) = [ t:ys | ys <- perm (hs ++ ts) ]

    splits []     = [([],[])]

    splits (x:xs) = ([],x:xs) : [ (x:ys,zs) | (ys,zs) <- splits xs ]

    rperm _  0 = [[]]

    rperm [] _ = []

    rperm xs n = [ x:ys | x <-xs, ys <- rperm xs (n-1) ]

　ticketが作るべき数 nを与えられた数字のリスト dsから四則を使って構成し表示する関数である．allterms

で可能なすべての計算木を構成しその中から評価すると nと等しい有理数になるものだけを濾過 (filter)し，それ

が空リストなら構成できなかった旨の文字列を，空ではないリストなら，その先頭を文字列に変換したものとなる．
permは与えられた数字のリストの順列を構成する関数，rpermは与えられた演算子を重複を許して n個使う順列を

構成する．これらの定義を second.hsに追加して ghciにロードすると実行できる．

    *Main> :load second.hs

    Compiling Main             ( second.hs, iterpreted )

    Ok, modules loaded: Main.

    *Main> ticket 10 "1158"

    (8/(1-(1/5)))

    *Main> ticket 24 "1346"

    (6/(1-(3/4)))

    *Main>

　これで切符問題を解くプログラムが書けた．Haskellを使ったプログラミングではデータや関数の在り様を記述す

るだけでよいという雰囲気を味わってもらえただろうか．

　今回のコードは説明しやすいように書いたので，実際に利用するプログラムとしては改良すべき点がいくつかある．

これらの一部を練習問題としておく．

練習問題 2　ticket関数は，解が存在しないときにエラーになる．これを変更して解が存在しないときはその旨の

メッセージを表示するようにせよ．

練習問題 3　与えられた数字に重複があるときには，permが同じ並びを複数回構成する．このため，計算木を重複

してチェックすることになる．数字の順列の構成部分を改良し，計算木のチェックの重複を除去せよ．

練習問題 4　2パスになっている treesを 1パスにせよ．すなわち，数字のリストから構成した仮の計算木をたど

りながら，最終的な計算木を構成し直すのではなく，一度で演算子も割り当てた計算木を構成するように改造

せよ．

練習問題 5　計算木が構成されると同時にその計算木の評価もなされるように，上で改造した treesに評価のパス

も組み込め．

注）連載「Haskellプログラミング」に掲載されたプログラムは次の URLから取ることができる．

　　http://www.sampou.org/haskell/ipsj/

